

Solution to Exercise 3: Derived fields and Check file commands unrelated to a specific field

Key Point(s):

- Commands that are specific for a field should be written by accessing the 3. Checks tab in the process bar as there they are checked for the accuracy of the grammar
- For commands that pertain to the file or the record rather than a field, open the Check file to make the necessary modifications. Note that the grammar is not checked here (errors will be reported when trying to open the data file for data entry).

Tasks:

- o Complete the revised A_EX03 .CHK file and verify your QES-REC-CHK triplet
- o Test the system by trying to enter data (4. Enter data)

Solution:

This is the Questionnaire file A_EX03 .QES:

This is the questionnaire for the laboratory register

labcode	Code of the laboratory	<input type="text"/>	
idcode	Unique identifier	<input type="text"/>	
labname	Name of the laboratory	<input type="text"/>	
serno	Laboratory serial number	<input type="text"/>	Enter 9001, 9002, ... if serial number is not unique and write data entry note (F5)
regdate	Registration date	<input type="text"/>	Enter 01/01/1800 if missing
sex	Examinee's sex	<input type="text"/>	
age	Examinee's age in years	<input type="text"/>	Enter 999 if missing
reason	Examination reason	<input type="text"/>	
res1	Result of specimen 1	<input type="text"/>	
res1sc	Result of specimen 1 scanty	<input type="text"/>	
res2	Result of specimen 2	<input type="text"/>	
res2sc	Result of specimen 2 scanty	<input type="text"/>	
res3	Result of specimen 3	<input type="text"/>	
res3sc	Result of specimen 3 scanty	<input type="text"/>	

This is the Check file A_EX03 .CHK:

```
LABELBLOCK
  LABEL label_labname
    Awuna ML_J
  END
  LABEL label_sex
    1 Female
    2 Male
```

```

    9 "Sex not recorded"
END
LABEL label_reason
    0 Diagnosis
    1 "Follow-up at 1 month"
    2 "Follow-up at 2 months"
    3 "Follow-up at 3 months"
    4 "Follow-up at 4 months"
    5 "Follow-up at 5 months"
    6 "Follow-up at 6 months"
    7 "Follow-up at 7 months or later"
    8 "Follow-up, month not stated"
    9 "Reason not recorded"
END
LABEL label_result
    0 Negative
    1 "1+ positive"
    2 "2+ positive"
    3 "3+ positive"
    4 "Positive, not quantified"
    5 "Scanty, not quantified"
    6 "Scanty, quantified"
    9 "Result not recorded"
END
LABEL label_scanty
    0 "Not applicable"
    1 "1 AFB per 100 OIF"
    2 "2 AFB per 100 OIF"
    3 "3 AFB per 100 OIF"
    4 "4 AFB per 100 OIF"
    5 "5 AFB per 100 OIF"
    6 "6 AFB per 100 OIF"
    7 "7 AFB per 100 OIF"
    8 "8 AFB per 100 OIF"
    9 "9 AFB per 100 OIF"
END
END

BEFORE FILE
    DEFINE sernoTemp ____ GLOBAL
END

AFTER RECORD
    IF idcode=. THEN
        HELP "You cannot save a record without an identifier\n Please enter
all required information" TYPE=WARNING
        GOTO labname
    ENDIF
END

labcode
    NOENTER
END

idcode
    KEY UNIQUE 1
    NOENTER
END

labname
    COMMENT LEGAL USE label_labname SHOW

```

```

MUSTENTER
REPEAT
TYPE COMMENT labcode
END

serno
MUSTENTER
AFTER ENTRY
    sernoTemp=serno
    IF serno<1000 THEN
        sernoTemp="0"+serno
    ENDIF
    IF serno<100 THEN
        sernoTemp="00"+serno
    ENDIF
    IF serno<10 THEN
        sernoTemp="000"+serno
    ENDIF
END
END

regdate
RANGE 01/01/2000 31/12/2005
LEGAL
    01/01/1800
END
MUSTENTER
AFTER ENTRY
    idcode=labcode+"-"+year(regdate)+"-"+sernoTemp
END
END

sex
COMMENT LEGAL USE label_sex SHOW
MUSTENTER
TYPE COMMENT
END

age
RANGE 0 125
LEGAL
    999
END
MUSTENTER
END

reason
COMMENT LEGAL USE label_reason SHOW
MUSTENTER
TYPE COMMENT
END

res1
COMMENT LEGAL USE label_result SHOW
MUSTENTER
TYPE COMMENT
AFTER ENTRY
    IF res1<>6 THEN
        res1sc=0
        GOTO res2
    ENDIF

```

```

END
END

res1sc
COMMENT LEGAL USE label_scanty SHOW
MUSTENTER
TYPE COMMENT
AFTER ENTRY
    IF (res1=6) AND (res1sc=0) THEN
        HELP "Values of res1 and res1sc not compatible. Please verify"
        GOTO res1
    ENDIF
END
END

res2
COMMENT LEGAL USE label_result SHOW
MUSTENTER
TYPE COMMENT
AFTER ENTRY
    IF res2<>6 THEN
        res2sc=0
        GOTO res3
    ENDIF
END
END

res2sc
COMMENT LEGAL USE label_scanty SHOW
MUSTENTER
TYPE COMMENT
AFTER ENTRY
    IF (res2=6) AND (res2sc=0) THEN
        HELP "Values of res2 and res2sc not compatible. Please verify"
        GOTO res2
    ENDIF
END
END

res3
COMMENT LEGAL USE label_result SHOW
MUSTENTER
TYPE COMMENT
AFTER ENTRY
    IF res3<>6 THEN
        res3sc=0
        GOTO WRITE
    ENDIF
END
END

res3sc
COMMENT LEGAL USE label_scanty SHOW
MUSTENTER
TYPE COMMENT
AFTER ENTRY
    IF (res3=6) AND (res3sc=0) THEN
        HELP "Values of res3 and res3sc not compatible. Please verify"
        GOTO res3
    ENDIF
END
END

```

END

This is a completed data entry form:

This is the questionnaire for the laboratory register

labcode	Code of the laboratory	<input type="text" value="ML_J"/>
idcode	Unique identifier	<input type="text" value="ML_J-2002-0023"/>
labname	Name of the laboratory	<input type="text" value="Awuna"/>
serno	Laboratory serial number	<input type="text" value="23"/> Enter 9001, 9002, ... if serial number is not unique and write data entry note (F5)
regdate	Registration date	<input type="text" value="20/03/2002"/> Enter 01/01/1800 if missing
sex	Examinee's sex	<input type="text" value="1"/> Female
age	Examinee's age in years	<input type="text" value="23"/> Enter 999 if missing
reason	Examination reason	<input type="text" value="0"/> Diagnosis
res1	Result of specimen 1	<input type="text" value="6"/> Scanty, quantified
res1sc	Result of specimen 1 scanty	<input type="text" value="4"/> 4 AFB per 100 OIF
res2	Result of specimen 2	<input type="text" value="9"/> Result not recorded
res2sc	Result of specimen 2 scanty	<input type="text" value="0"/>
res3	Result of specimen 3	<input type="text" value="9"/> Result not recorded
res3sc	Result of specimen 3 scanty	<input type="text" value="0"/> Not applicable